

A SAS® Macro to Generate Information Rich Kaplan-Meier Plots

Chia-Ling (Ally) Wu, Seattle Genetics, Seattle, WA

ABSTRACT

The number of patients at risk at each time point in a Kaplan-Meier (KM) plot is critical information needed to interpret the curve. This paper describes a SAS® macro which provides a tabular summary of data used to generate the KM plot, including the number of subjects at risk, the number of events and total subjects, median survival rate and Cox proportional hazard ratio. This macro also supports a customized output header, labels for time interval (X-axis) and survival rate (Y-axis), reference line, and tick marks for time interval (X-axis). In addition, the ability to add a filter to the data to generate KM plots by various subsets is provided. The output destination can be PDF, RTF or html files.

This macro applies the SAS® PROC LIFETEST procedure to delineate Kaplan-Meier survival plots and computes the survival function by log-rank test and Wilcoxon test. This macro also uses the PROC PHREG procedure based on the Cox proportional regression model to estimate the effect of predictors on hazard rates. With SAS® ODS SELECT tables produced by PROC LIFETEST and PROC PHREG can create an output dataset, which contains statistical information for subsequent use.

INTRODUCTION

Time to event is one of the most often used analysis in the oncology field. In time-to-event data, investigators follow subjects until they reach a pre-specified endpoint (for example, death or progression of disease). However, subjects sometimes withdraw from a study, or the study is completed before the endpoint is reached. In these cases, the survival times are censored. The uncensored survival times are referred to as event times.

In SAS®, the PROC LIFETEST procedure creates a Kaplan-Meier survival plot, computes the survival function from lifetime data, and compares the survivor function between groups by log-rank test and Wilcoxon test. The PROC PHREG procedure is used in Cox proportional regression model to estimate the effect of predictors on hazard rates. SAS® ODS SELECT in PROC LIFETEST and PROC PHREG provide statistical information in the datasets for further analysis use.

All the examples in this macro use the Bone Marrow Transplant Patients data <*Sashelp.bmt*> from SAS library. At the time of transplant, each patient is classified into one of three risk categories: ALL (acute lymphoblastic leukemia), AML-Low Risk (acute myelocytic leukemia, low risk), and AML-High Risk, which are defined in the GROUP variable. The endpoint of interest is the disease-free survival time, which is the time in days to death, relapse, or the end of the study, which is defined in the STATUS variable (1 is an event and 0 is censored). In order to demonstrate, I subset the AML subjects only, coded GROUP as 1 for AML-Low Risk, 2 for AML-High Risk, and created a dummy AGE variable.

OVERVIEW OF MACRO

The SAS® macro, *mKMplot*, in this paper aims to create a customized Kaplan-Meier plot with a table for subjects at risk and optional estimates which include survival rate with 95% CI, Cox proportional hazard ratio with 95% CI and the corresponding p-value. This macro also provides a customized output, such as header of figure, label for time interval (X-axis) and survival rate (Y-axis), and tick marks at desired time points for the time interval (X-axis) and subjects at risk table. In addition, a filter for the data is included to allow subsetting. The output destination can be PDF, RTF or html files from SAS® ODS.

Figure 1 is an example output generated by *mKMplot* macro consisting of four parts: Kaplan-Meier plot, the table for the number of subjects at risk, the event numbers, and the Cox proportional hazard ratio. These macro parameters are described in Table 1. The number captions identify the optional parameters in the macro, and a default value will be provided if they are left blank. The four components of the KM plot are defined below.

1. Kaplan-Meier Plot: PROC LIFETEST delineates the Kaplan-Meier plot. The user can define labels of X-axis and Y-axis, tick marks at desired time points, and whether to include a reference line at the median level. If the analysis is done by different groups, the user can include a macro variable `_STRATAVAR` in the STRATA statement. Then, the macro expects that the dataset contains both the code and the decode variables in the form "VAR" and "VAR_C". In the example below, the dataset should have variable GROUP and GROUP_C. The numeric variable, GROUP, would have value as 1 and 2, while the character variable, GROUP_C, would have values AML-Low Risk and AML-High Risk respectively.
2. Subjects at Risk: The user can define label of table header and subject numbers calculated at desired time

points, which are the same as the tick marks in the KM plot. The table for number of subjects at risk is provided by option ATRISK from PROC LIFETEST.

3. Subject/ Event Numbers: The legends for strata, event number and total subject number are default outputs, whereas median survival rate with 95% CI are optional outputs.
4. The Cox proportional Hazard Ratio: The Cox proportional hazard ratio, 95% CI and p-value are optional outputs. The user can output the hazard ratio (with 95% CI) only, hazard ratio with p-value, or neither of these estimates. In this macro, using PROC PHREG provides the Cox proportional regression estimates using the EXACT method, which is the most accurate method for the continuous time-scale model but also takes more computational resource. The hazard ratio is calculated based on the reference group specified in the macro, or the default value is like SAS default using the last sorted group if not specified. The user should refer to the study statistical analysis plan to verify whether or not the EXACT method is the appropriate method to use.

④ Figure 1: Kaplan-Meier Plot of Disease Free Survival for Transplant Patients




Figure 1 shows disease free survival among AML-Low Risk and High Risk subjects who had a bone marrow transplant based on the Bone Marrow Transplant Patients data <SasHELP.BMT> from SAS library. The event of interest is disease relapse including death. The censor variable will be 1 if the subject relapsed or died during the study and 0 if the subject remained disease free or is lost to follow up. The user also has to specify the time variable to be used in the PROC LIFETEST and PROC PHREG model. In this example, the survival estimates are based on AML risk stratification and AML low risk (uncoded value 1) as reference group. This figure presents number of subjects at risk and tick marks for every 10 months. Optional estimates of median survival rate, hazard ratio and p-value are presented. If the median survival rate cannot be obtained since more than 50% of subjects are free from disease relapse, it is presented as "NA", like the upper level of the median survival rate in this case. Figure 1 does not use macro variable _STRATALIS, which is applied in the STRATA statement in PROC PHREG. If we add, for example AGE, the hazard ratio will calculate by age group stratification, which is a common way that is used to control covariates.

```

%mkMplot ( _indata= BMT
           ,_timevar= T_M
           ,_censorvar= STATUS(0)
           ,_stratavar= GROUP
           ,_tinterval= 10
           ,_ref= 1
           ,_stratalis=
           ,_datafl=
           ,_header= "Figure 1: Kaplan-Meier Plot of Disease Free Survival for Transplant
                     Patients"
           ,_Xlabel= "Time (Month)"
           ,_Ylabel= "Disease Free"
           ,_ARTitle="Number of Subjects at Risk"
           ,_Refline=Y
           ,_fnmed= Y
           ,_fnhr= Y
           ,_fnp= Y
           ,_outtype=rtf);


```

MACRO VARIABLES

Table 1. Definition of the parameters and optional estimates.

	Macro Variable	Required	Description of Function
	_INDATA	Y	Input dataset assumed to be one record per subject.
	_TIMEVAR	Y	Time at risk variable.
	_CENSORVAR	Y	Censoring variable should include the value to identify a censored observation. For instance, STATUS(0) identifies 0 as censored data and 1 as the event of interest.
①	_STRATAVAR		Analysis effect for KM analysis and the Cox regression. For example, AML Risk in this case, requires numeric format.
②	_REF		Study reference group. For example, 1 (AML-Low Risk) in this case. Default is the last sorted value in the dataset.
②	_STRATALIS		Analysis stratification for the Cox regression.
	_DATAFL		Select partial data to do survival analysis. Acts in the WHERE statement in the PROC LIFETEST and PROC PHREG procedure, should be used in SAS logic, for example, AGE>45.
③	_TINTERVAL		Time interval for displaying tick marks at desired time points for X-axis and subjects at risk table.
④	_HEADER		The header of the figure.
⑤	_XLABEL		The label of X-axis (Time interval).
⑥	_YLABEL		The label of Y-axis (Survival Rate).
⑦	_ARTITLE		The label of Number of subjects at risk.
⑧	_REFLINE		Present reference line for median survival rate. Default value is N.
⑨	_FNMED		The footnote shows survival median or not. Default value is N.
⑩	_FNHR		The footnote shows hazard ratio or not. Default value is N.
⑪	_FNP		The footnote shows p-value for hazard ratio or not. Only functions when hazard ratio presents. Default value is N.
	_OUTTYPE		Output document type. Default is RTF file.

Figure 2: Kaplan-Meier Plot of Disease Free Survival for Transplant Patients Age >45 Years



Number of Subjects at Risk									
AML-Low Risk	54	42	33	24	20	11	6	1	0
AML-High Risk	45	18	11	10	7	6	3	1	0

— AML-Low Risk Events :25/54, Median :73.5(21.4,NA)
- - - AML-High Risk Events :34/45, Median :6.1(3.8,13.0)

Hazard ratio (reference=AML-High Risk) : 0.39 (0.23,0.66), p-value : <0.001

Figure 2 is another example of disease free survival among AML-Low Risk and High Risk subjects who received a bone marrow transplant and are 45 years or older. The macro variable DATAFL acts in the WHERE statement to subset the data accordingly. In this case, without specifying the reference group, this macro uses default value last sorted order as the reference group, which is GROUP=2 and presented in coded AML-High Risk. Therefore, we can see the hazard ratio is almost reverse from Figure 1 (i.e. the slightly different from reverse number was due to the age subset). The time interval is presented for every 12 months. The reference line for the median survival rate is not presented. Aside from the subjects at risk and events number, this figure also presents median survival rate and Cox proportional hazard ratio with the associated 95% CI and p-value.

```

%MKMplot ( _indata= BMT
            ,_timevar= T_M
            ,_censorvar= STATUS(0)
            ,_stratavar= GROUP
            ,_tinterval= 12
            ,_ref=
            ,_stratalis=
            ,_datafl= AGE >45
            ,_header= "Figure 2: Kaplan-Meier Plot of Disease Free Survival for Transplant
                      Patients Age >45 Years"
            ,_Xlabel= "Time (Month)"
            ,_Ylabel= "Disease Free"
            ,_ARtitle="Number of Subjects at Risk"
            ,_Refline= N
            ,_fnmed= Y
            ,_fnhr= Y
            ,_fnp= Y
            ,_outtype=rtf);
    
```

Figure 3: Kaplan-Meier Plot of Disease Free Survival for Transplant Patients




Figure 4: Kaplan-Meier Plot of Disease Free Survival for Transplant Patients among AML Risks




Figure 3 and Figure 4 are examples when optional settings are not specified; the only difference is Figure 4 has specified a stratification by level of risk. Therefore, we can see that Figure 3 presents overall disease free survival and event numbers whereas in Figure 4 the results are calculated by AML risk. The label of the X-axis is based on the time variable label and the label of Y-axis is a SAS default value. The footnote for the median survival rate, Cox proportional hazard ratio and p-value are not shown in the figure since the default value for those are not to present.

```
%mKMplot ( _indata= BMT
            ,_censorvar= STATUS(0)
            ,_timevar= T_M);

%mKMplot ( _indata= BMT
            ,_censorvar= STATUS(0)
            ,_timevar= T_M
            ,_stratavar=GROUP);
```

SAS CODE

```
/*
Data requirements:
  1. Input dataset assumed to be one record per subject.
  2. Stratification (macro var _stratavar) needs a numeric variable.
   Data need another variable <_stratavar>_c has corresponding label for
   proc format. e.g. TRTPN/TRTPN_C: 1/Drug A, 2/Placebo.
*/

%macro mKMplot ( _indata=          /*Required.Input dataset assumed to be one record per
                                     subject*/
                ,_timevar=         /*Required. Time variable*/
                ,_censorvar=       /*Required. Censoring variable, should include value
                                     to identify a censored observation e.g: HYPER(0),
                                     0 is censor*/
                ,_stratavar=       /*Permissible. Analysis stratification for KM,
                                     usually treatment group.*/
                ,_stratalis=       /*Permissible. Study stratification groups*/
                ,_datafl=         /*Permissible. Data filter for LIFETEST and PHREG
                                     (where=&datafl)), please use SAS logic. e.g.
                                     GROUP=1*/
                ,_tinterval=       /*Permissible. Time interval for displaying.
                                     Default =3*/
                ,_header=         /*Permissible. The header of the figure*/
                ,_Xlabel=         /*Permissible. The label of X-axis (Time interval)*/
                ,_Ylabel=         /*Permissible. The label of Y-axis (Survival)*/
                ,_ARTitle=        /*Permissible. The label of Number of subjects at
                                     risk*/
                ,_Refline=        /*Permissible. Present reference line for median
                                     survival rate. Default=N.*/
                ,_ref=            /*Permissible. Study reference group*/
                ,_fnmed=          /*Permissible. The footnote shows survival median or
                                     not. Default=N.*/
                ,_fnhr=          /*Permissible. The footnote shows hazard ratio or
                                     not. Default=N.*/
                ,_fnp=           /*Permissible. The footnote shows pvalue for hazard
                                     ratio or not. Default=N.*/
                ,_outtype= rtf     /*Permissible. Output document type. Default=rtf.*/ );

/*-----*/
/* ticks for X-axis */
/*-----*/
%if &_TINTERVAL = %then %let _TINTERVAL = 3; /*default=3*/

proc sql noprint;
  select &_TINTERVAL.*ceil((max(&_TIMEVAR.)/&_TINTERVAL.)) into: _maxi
  from &_INDATA.
  ;
run;
quit;

/*-----*/
/* strata */
/*-----*/
%if &_STRATAVAR.= %then %do;
  data &_INDATA.;
    set &_INDATA.;
    DMY = 1;
    DMY_C = "All Subjects";
  run;
  %let _STRATAVAR = DMY;
%end;

/*-----*/
/* label for strata: STRATA */
/*-----*/
proc sort data= &_INDATA out = fmt nodupkey;
  by &_STRATAVAR.;
```

```

run;

%let _STRATAVAR_C = %sysfunc(cats(%scan(&_STRATAVAR.,1),_C));%put &_STRATAVAR_C;

data fmt (keep=fmtname type start label &_STRATAVAR.);
  set fmt end=eos;
  by &_STRATAVAR.;

  retain fmtname "STRATA" type "N";
  length start end $8. label $40.;

  start = strip(put(_N_,best.));
  end = strip(put(_N_,best.));
  label = &_STRATAVAR_C.;

  if eos then call symputx ('_Nstrata', _N_);
run;

proc format library=work cntlin=fmt;
run;

/*-----*/
/*PROC LIFETEST for KM-plot, subject at risk, survival rate median, event numbers*/
/*-----*/
ods graphics / reset;
ods exclude all;
ods trace on;

ods output survivalplot=zz_plotdata CensoredSummary=zz_esum Quartiles=zz_qtp
  %if &_Nstrata.>= 2 %then HomTests=zz_hstest ;;
proc lifetest data=_INDATA %if %bquote(&_DATAFL.) ne %then %do; %str((where=(&_DATAFL.))) %end;
  method=pl
  plots=survival(atrisk(outside)=0 to &_maxi. by &_TINTERVAL.);;
  ods select SurvivalPlot CensoredSummary Quartiles %if &_Nstrata.>= 2 %then Homtests;;
  time &_TIMEVAR. * &_CENSORVAR.;
  strata &_STRATAVAR.;;
run;

data zz_plotdata;
  set zz_plotdata;
  by STRATUM;
  retain CUMEVENT SURVPROB;
  if first.STRATUM then do;
    CUMEVENT=0;
    SURVPROB=1;
  end;
  else do;
    if EVENT ^= . then CUMEVENT=CUMEVENT+EVENT;
    if SURVIVAL ^= . then SURVPROB=SURVIVAL;
  end;
run;

/*-----*/
/* Data for plot */
/*-----*/
data fSURVIVAL;
  set zz_plotdata;
  format STRATUMNUM strata.;
run;

/*-----*/
/* footnote event/ median */
/*-----*/
proc sql;
  create table fnl_0 as
  select a._STRATAVAR., a.STRATUM, a.TOTAL, a.FAILED, b.ESTIMATE, b.LOWERLIMIT, b.UPPERLIMIT
  from zz_esum as a
  left join zz_qtp (where=(PERCENT=50)) as b
  on a._STRATAVAR.=b._STRATAVAR.
  having ^missing(&_STRATAVAR.)

```

```

order by STRATUM
;
quit;

%if &_REF= %then %do;
data _null_;
set fmt end=eos;
by START;
if eos then call symputx ('_REF', input(&_STRATAVAR.,best.));
run;
%end;

proc sql;
select LABEL into: _REFGP
from fmt
where &_STRATAVAR. = &_REF
;
quit;

data fn1;
set fn1_0;
by STRATUM;

/*length setting- for output space*/
%if &_fnmed=Y %then %do; length x $100.; %end;
%else %do; length x $120.; %end;

if missing(TOTAL) then TOTAL=0;
if missing(FAILED) then FAILED=0;

array zero1(*) ESTIMATE LOWERLIMIT UPPERLIMIT;
array zero2(*) $8. ESTIMATE_ LOWERLIMIT_ UPPERLIMIT_;
do i=1 to dim(zero1);
if missing(zero1(i)) then zero2(i)="NA";
else zero2(i) = strip(put(zero1(i),5.1));
end;

x1 = cats("Events : ", put(FAILED,best.) , "/", put(TOTAL,best.));
x2 = cats("Median : ", ESTIMATE_, " (", LOWERLIMIT_, ",", UPPERLIMIT_, ")" );
%if &_fnmed=Y %then %do; x = catx(" ", x1, x2); %end;
%else %do; x = x1; %end;

if &_stratavar = &_ref then call symputx ('_FAILED', FAILED); /*Failed - event number of ref
group*/
run;

/* depends on stratification numbers */
%do i=1 %to &_Nstrata.;
%global _event_&i;
data _null_;
set fn1;
if _n_=&i. then call symput("_event_&i.",x);
run;
%end;

/*footnote HR and P, at least Straum >1*/
%if &_Nstrata.>= 2 %then %do;

ods output ParameterEstimates=zz_hazard;
proc phreg data=&_INDATA. %if %bquote(&_DATAFL.) ne %then %str((where=(&_DATAFL.)));;
ods select ParameterEstimates;
class &_STRATAVAR. (ref="&_REF.") &_STRATALIS.;
model &_TIMEVAR.* &_CENSORVAR.=&_STRATAVAR./ties=EXACT risklimits alpha=0.05 ;
strata &_STRATALIS.;
run;
ods exclude none;

proc sql;
create table fn2_0 as
select a.*,b.LABEL as _STRATANAME, monotonic() as _N_

```



```

        from zz_hazard as a
        left join FMT as b
            on input(a.CLASSVAL0,best.)=b.&_STRATAVAR.
    ;;
quit;

%do i=1 %to %eval(&Nstrata.-1);
    proc sql;
        select _STRATANAME into: _STRATANAME_&i
            from fn2_0
            where _N_ = &i.
        ;
    quit;
    %put &&_STRATANAME_&i ;
%end;

data fn2;
    set fn2_0;

    array zero1(*) HAZARDRATIO HRLOWERCL HRUPPERCL PROBCHISQ;
    array zero2(*) $8. HAZARDRATIO_ HRLOWERCL_ HRUPPERCL_ PROBCHISQ_;
    do k=1 to dim(zero1);
        %if &_FAILED. ^= 0 %then %do;
            if missing(zero1(k)) then zero2(k)="NA";
            else zero2(k) = strip(put(zero1(k),5.2));
        %end;
    end;

    if . < PROBCHISQ < 0.001 then PROBCHISQ_ = "<0.001";
    else if PROBCHISQ > 0.999 then PROBCHISQ_ = ">0.999";

    %if &_fnhr=Y %then %do;

        %do i=1 %to %eval(&Nstrata.-1);
            %global _hr1_&i _hr2_&i _hr3_&i _hr4_&i;
            if _n_ = &i then do;
                %if &_Nstrata. = 2 %then
                    %let _hr1_&i = %str(Hazard ratio (reference=%trim(&_REFGP.)) : );
                    %if &_Nstrata. > 2 %then
                        %let _hr1_&i = %str(Hazard ratio for %trim(&&_STRATANAME_&i)
(reference=%trim(&_REFGP.)) : );
                %if &_FAILED. = 0 %then %do;
                    %let _hr2_&i = {unicode '221e'x};
                    call symput ("_hr3_&i" , " (NA,NA)");
                %end;
                %else %do;
                    call symput ("_hr2_&i" , cats(HAZARDRATIO_));
                    call symput ("_hr3_&i" , " (" || cats(HRLOWERCL_ , ",",
HRUPPERCL_ , ")"));
                %end;
                %if &_fnp=Y %then call symput ("_hr4_&i" , cats(", p-value : ", PROBCHISQ_));
            end;
            %put &&_hr1_&i &&_hr2_&i &&_hr3_&i &&_hr4_&i;
        %end;
    %end;

run;

%end;

/*-----*/
/*      for output      */
/*-----*/

%let _lev2=%sysevalf((%sysevalf(&Nstrata)*.04)+.01);
%let _lev3=%sysevalf(&_lev2);
%let _lev4=%sysevalf((%sysevalf(&Nstrata-1)*.04)+.01);
%let _lev1=%sysevalf(1-&_lev2-&_lev3-&_lev4);

%put &_lev1 &_lev2 &_lev3 &_lev4;

```

```

%macro temp;

proc template;
  define statgraph f_SURVIVAL;
    dynamic TOTAL FAILED ;
    mvar _Nstrata _ARtitle _Ylabel _Xlabel _header _lev1 _lev2 _lev3 _lev4;

    begingraph / border=FALSE designwidth=20cm designheight=14cm;

    layout lattice / rows=4 rowweights=(&_lev1 &_lev2 &_lev3 &_lev4) rowgutter=0
    columndatarange=union ;

    /*1st cell- KM plot*/
    cell;
    layout overlay / border=false
      yaxisopts=(offsetmin=0 linearopts=(thresholdmin=0 viewmin=0
        tickvaluesequence=(start=0 end=1 increment=.1))
        %if &_Ylabel^= %then %do; label=&_Ylabel. labelattrs=(size=9pt) %end;)
      xaxisopts=(offsetmin=0 display=(line label ticks tickvalues)
        %if &_Xlabel^= %then %do; label=&_Xlabel. labelattrs=(size=9pt) %end;
        linearopts=(thresholdmin=0 tickvaluesequence=(start=0 end=%eval(&_maxi)
increment=%eval(&_tinterval) )));
      scatterplot x=TIME y=CENSORED / group=STRATUMNUM name="scat";
      stepplot x=TIME y=SURVIVAL / group=STRATUMNUM name="step";
      %if &_Refline=Y %then %do; referenceline y=0.5 / lineattrs=(color=lightgray); %end;
    endlayout;
    endcell;

    /*2nd cell- AT Risk*/
    cell;
    %if &_ARtitle^= %then %do;
    cellheader;
      entry halign=left textattrs=(size=8pt weight=bold) &_ARtitle. / border=FALSE;
    endcellheader;
    %end;
    layout overlay/ walldisplay=none xaxisopts=(display=none);
      blockplot x=TATRISK block=ATRISK / class=STRATUMNUM blockindex=STRATUMNUM
        repeatedvalues=true display=(label values)
        valuehalign=start valuefitpolicy=truncate
valueattrs=graphdatatext(size=8pt)
        labelposition=left labelattrs=graphvaluetext
labelattrs=(size=8pt) includemissingclass=false;
    endlayout;
    endcell;

    /*3rd cell- Event Number*/
    cell;
    layout overlay/ walldisplay=none xaxisopts=(display=none) pad=(top=0 left=5 bottom=0);
    discretelegend "step" / valueattrs=(size=8pt) border=false displayclipped=true halign=left;
    layout gridded / rows=&_Nstrata. border=false;
      %do i=1 %to &_Nstrata.;
        entry halign=left textattrs=(size=8pt) "%nrbquote(&&_event_&i)" ;
      %end;
    endlayout;
    endlayout;
    endcell;

    /*4th cell- Hazard Ratio*/
    %if &_Nstrata. >=2 %then %do;
    cell;
    layout overlay/ walldisplay=none xaxisopts=(display=none);
      layout gridded / rows=%eval(&_Nstrata.-1) border=false halign=left;
        %do i=1 %to %eval(&_Nstrata.-1);
          %if &_FAILED. ^= 0 %then %do; entry halign=left textattrs=(size=8pt) " " "&&_hr1_&i"
"&&_hr2_&i" "&&_hr3_&i" "&&_hr4_&i"; %end;
          %else %if &_FAILED. = 0 %then %do; entry halign=left textattrs=(size=8pt) " "
"&&_hr1_&i" "&&_hr2_&i" "&&_hr3_&i" "&&_hr4_&i"; %end;
        %end;
    endlayout;
    endlayout;
    endcell;
  enddefine;
%macroend;

```

```

        %end;

        endlayout;
        endgraph;
        end;
run;
quit;

%mend temp;
%temp;

options orientation=landscape papersize=letter nodate nonumber;
ods noproctitle escapechar="`";
ods listing style=listing;

ods graphics / reset noborder imagename="fSURVIVAL" ;
ods listing gpath="&_outputPath\";
ods &_OutType file ="&_outputPath\survival.&_outtype"
    style=listing nogfootnote nogtitle bodytitle toc_data headery=720 footery=720;

proc sgrender data=fSURVIVAL template=f_SURVIVAL
    %if &_HEADER^= %then description=&_HEADER.;;
run;

ods &_OutType close;
ods listing ;

%mend mKMplot;

```

CONCLUSION

This paper presents a SAS® macro to generate a survival plot with customized estimates and output settings providing an efficient creation of Kaplan-Meier output for time-to-event analyses.

REFERENCES

- Berglund, P. A. 2011. "An Overview of Survival Analysis using Complex Sample Data", presented at SAS Global Forum, Las Vegas, NV, 2011.
- Gharibvand, L. 2008. "A Step-by-Step Guide to Survival Analysis", presented at SAS WUSS Forum, Universal City, CA, 2008.
- Kaplan, E. L. and Meier, P. (1958), "Nonparametric Estimation from Incomplete Observations," Journal of the American Statistical Association, 53, 457–481.
- Kuhfeld, W. F. and So, Y. 2013. "Creating and Customizing the Kaplan-Meier Survival Plot in PROC LIFETEST", presented at SAS Global Forum, San Francisco, CA, 2013.
- UCLA Institute for Digital Research and Education. <http://stats.idre.ucla.edu/sas/seminars/sas-survival/>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Chia-Ling (Ally) Wu
 Seattle Genetics
awu@seagen.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.